

# A Datalog Primer

Jef Wijsen

March 19, 2020

## Abstract

This document introduces the syntax and semantics of the language “*datalog with stratified negation*,” and shows that programs in this language can be executed in polynomial time. After studying this document, students should be able to write their own programs in this language.

## 1 Schema and Database Instance

Datalog is a query language for relational databases. A *relational database schema* (or *schema* for short) is a finite set of relation names with associated arities. We write  $R/n$  to indicate that  $R$  is a relation name of arity  $n$ .

**Example 1.1.** In our running example, we will use a schema that contains  $Knows/2$  and  $Owens/2$ .  $\square$

A *fact* is an expression  $R(c_1, \dots, c_n)$  where each  $c_i$  is a constant. A *database instance*  $I$  is a finite set of facts.

**Example 1.2.** The set  $I = \{Knows(Jeb, Don), Knows(Don, Jeb), Knows(An, Don), Knows(Ed, An), Owens(Don, iPad), Owens(Don, iPod), Owens(Jeb, iPod)\}$  is a database instance.  $\square$

The term *EDB predicate* is often used as a synonym for “relation name of the relational database schema.” The letter  $E$  in *EDB* stands for *Extensional*. I like to think about it as *Existential*: that what really *Exists* in the database.

**Example 1.3.**  $Knows$  and  $Owens$  are EDB predicates.  $\square$

As a minor comment, in mathematical logic, one often uses the term *predicate symbol* for what I call a *predicate* in this document.

## 2 Datalog Syntax

We assume a set of *IDB predicates* with associated arities. The IDB predicates are distinct from the EDB predicates. The letter  $I$  in *IDB* stands for *Intensional*. I like to think about it as *Inferred*, because datalog is all about *Inferring* IDB facts from a database instance of EDB facts.

A fact  $P(c_1, \dots, c_n)$  is an *EDB fact* if  $P$  is an EDB predicate, and is an *IDB fact* if  $P$  is an IDB predicate. An *atom* is an expression  $P(t_1, \dots, t_n)$  where each  $t_i$  is a constant or a variable. The letter  $t$  refers to *term*, which is a variable or a constant.

A *datalog rule* is an expression

$$\underbrace{H(\vec{t}_0)}_{\text{rule head}} \leftarrow \underbrace{R_1(\vec{t}_1), \dots, R_\ell(\vec{t}_\ell), \neg R_{\ell+1}(\vec{t}_{\ell+1}), \dots, \neg R_m(\vec{t}_m)}_{\text{rule body}} \quad (1)$$

where

- $H$  is an IDB predicate,
- each  $R_i$  is an EDB predicate or an IDB predicate, and
- each  $\vec{t}_i$  is a sequence of terms (of appropriate length).

Note that the same predicate can occur more than once in a rule. Within a rule, we distinguish between the *head* and the *body*, as indicated in (1). Every rule must be *safe*:

**Safety Requirement:** Every variable that occurs in a rule (in its head or its body), must occur in a nonnegated atom of the rule body.

Informally, the safety requirement guarantees that every variable can be restricted to a finite domain.

Significantly, EDB predicates cannot be used in rule heads. A *datalog program*  $\mathcal{P}$  is a finite set of rules, which can be written in any order.

**Example 2.1.** Assume IDB predicates *Happy/1* and *Unhappy/1*.

$$\begin{aligned} \text{Happy}(x) &\leftarrow \text{Owns}(x, \text{iPad}), \text{Owns}(x, \text{iPod}) \\ \text{Happy}(x) &\leftarrow \text{Knows}(x, y), \text{Happy}(y) \\ \text{Happy}(x) &\leftarrow \text{Knows}(x, y), \neg \text{Knows}(x, \text{Don}) \\ \text{Unhappy}(x) &\leftarrow \text{Owns}(x, y), \neg \text{Happy}(x) \end{aligned}$$

It can be easily seen that all rules are safe. In the last rule, the variable  $x$  is restricted to the constants that occur at the first position of some *Owns*-fact.  $\square$

Informally, the semantics is as follows. An *instantiation* of a rule is obtained by replacing all variables with constants. Such an instantiation is called a *ground rule*. Then, whenever the body of a ground rule is true, we infer that its rule head is also true. The body of a ground rule is true if every nonnegated fact is true, and every negated fact is false.

**Example 2.2.** Consider the program of Example 2.1. An instantiation of the first rule is:

$$\text{Happy}(\text{Don}) \leftarrow \text{Owns}(\text{Don}, \text{iPad}), \text{Owns}(\text{Don}, \text{iPod})$$

Since *Owns*(Don, iPad) and *Owns*(Don, iPod) are in the database instance, they are both true. Therefore, the body is true, and we can infer that *Happy*(Don) is true. An instantiation of the second rule is:

$$\text{Happy}(\text{Jeb}) \leftarrow \text{Knows}(\text{Jeb}, \text{Don}), \text{Happy}(\text{Don})$$

Since *Knows*(Jeb, Don) is in the database instance, and *Happy*(Don) has been inferred to be true, we can infer that *Happy*(Jeb) is true. An instantiation of the third rule is:

$$\text{Happy}(\text{Ed}) \leftarrow \text{Knows}(\text{Ed}, \text{An}), \neg \text{Knows}(\text{Ed}, \text{Don})$$

Since *Knows*(Ed, An) is in the database instance, and *Knows*(Ed, Don) is false (because it is not in the database instance), we can infer that *Happy*(Ed) is true.  $\square$

## 3 Stratified Negation

### 3.1 The Problem of “Circular” Negation

If  $P$  is an EDB predicate, then the truth of a fact  $P(c_1, \dots, c_n)$  is naturally defined as follows:

- $P(c_1, \dots, c_n)$  is true if it is in the database instance; and
- $P(c_1, \dots, c_n)$  is false (and therefore  $\neg P(c_1, \dots, c_n)$  is true) if  $P(c_1, \dots, c_n)$  is not in the database instance.

This way of defining truth is also called the *Closed World Assumption (CWA)*.

If  $P$  is an IDB predicate, then it is trickier to determine the truth of a fact  $P(c_1, \dots, c_n)$ . Consider the following program:

$$\begin{aligned} \text{Man}(x) &\leftarrow \text{Owns}(x, y), \neg \text{Female}(x) \\ \text{Female}(x) &\leftarrow \text{Owns}(x, y), \neg \text{Man}(x) \end{aligned}$$

We can obtain the following ground rules:

$$\begin{aligned} \text{Man}(\text{Jeb}) &\leftarrow \text{Owns}(\text{Jeb}, \text{iPod}), \neg \text{Female}(\text{Jeb}) \\ \text{Female}(\text{Jeb}) &\leftarrow \text{Owns}(\text{Jeb}, \text{iPod}), \neg \text{Man}(\text{Jeb}) \end{aligned}$$

Note incidentally that every program without variables is necessarily safe. Since  $Owns(\mathbf{Jeb}, \mathbf{iPod})$  is true (because it is in the database), we can simplify these ground rules as follows:

$$\begin{aligned} Man(\mathbf{Jeb}) &\leftarrow \neg Female(\mathbf{Jeb}) \\ Female(\mathbf{Jeb}) &\leftarrow \neg Man(\mathbf{Jeb}) \end{aligned}$$

This program is cumbersome: the first rule allows us to infer  $Man(\mathbf{Jeb})$  if we cannot infer  $Female(\mathbf{Jeb})$ ; and the second rule allows us to infer  $Female(\mathbf{Jeb})$  if we cannot infer  $Man(\mathbf{Jeb})$ . This leads to an impossible situation: the first rule cannot be executed before the second rule, and the second rule cannot be executed before the first rule. Significantly, a datalog program does not depend on the order in which its rules are listed. So the previous program is really the same as:

$$\begin{aligned} Female(\mathbf{Jeb}) &\leftarrow \neg Man(\mathbf{Jeb}) \\ Man(\mathbf{Jeb}) &\leftarrow \neg Female(\mathbf{Jeb}) \end{aligned}$$

### 3.2 Stratified Negation

A *stratification* of a datalog program assigns a nonnegative integer, called *stratum*, to every IDB predicate of the program such that for every program rule

$$H(\vec{t}_0) \leftarrow R_1(\vec{t}_1), \dots, R_\ell(\vec{t}_\ell), \neg R_{\ell+1}(\vec{t}_{\ell+1}), \dots, \neg R_m(\vec{t}_m)$$

the following holds:

**Stratification Requirement:**

- for every  $i \in \{1, \dots, \ell\}$ , if  $R_i$  is an IDB predicate, then the stratum of  $R_i$  is smaller than or equal to the stratum of  $H$ ; and
- for every  $i \in \{\ell + 1, \dots, m\}$ , if  $R_i$  is an IDB predicate, then the stratum of  $R_i$  is strictly smaller than the stratum of  $H$ .

That is, if the stratum of the head predicate  $H$  is the integer  $h$ , then all IDB predicates in the rule body must have a stratum that is  $\leq h$ . Moreover, negated IDB predicates in the rule body must have a stratum  $< h$ .

Not all datalog programs have a stratification. In particular, we claim that there is no stratification for:

$$\begin{aligned} Man(x) &\leftarrow Owns(x, y), \neg Female(x) \\ Female(x) &\leftarrow Owns(x, y), \neg Man(x) \end{aligned}$$

To prove our claim, assume that we assign stratum  $m$  to  $Man$ , and stratum  $f$  to  $Female$ . Then, the Stratification Requirement applied to the first rule implies  $f < m$ . Likewise, the Stratification Requirement applied to the second rule implies  $m < f$ . But no stratification can possibly satisfy both  $f < m$  and  $m < f$ .

We say that a datalog program is *stratified* if it has a stratification. From now on, we will reject all programs that are not stratified. If we have a stratification, we will assume without loss of generality that the strata are numbered  $0, 1, 2, \dots$

**Example 3.1.** We compute a stratification for the program:

$$\begin{aligned} Happy(x) &\leftarrow Owns(x, \mathbf{iPad}), Owns(x, \mathbf{iPod}) \\ Happy(x) &\leftarrow Knows(x, y), Happy(y) \\ Happy(x) &\leftarrow Knows(x, y), \neg Knows(x, \mathbf{Don}) \\ Unhappy(x) &\leftarrow Owns(x, y), \neg Happy(x) \end{aligned}$$

The Stratification Requirement only talks about IDB predicates, and therefore we can ignore the EDB predicates  $Owns$  and  $Knows$ . Let the stratum of  $Happy$  be  $h$ , and the stratum of  $Unhappy$  be  $u$ . Because of the second program rule, we must have  $h \leq h$ , which is vacuously satisfied. Because of the last program rule, we must have  $h < u$ . We obtain a valid stratification by letting  $h = 0$  and  $u = 1$ . This stratification is summarized in the following table:

stratum 0: *Happy*  
stratum 1: *Unhappy*

□

### 3.3 Semantics of Stratified Negation

Assume we have computed a stratification for a stratified datalog program  $\mathcal{P}$ , and let  $m$  be the greatest stratum in our stratification. For every  $h \in \{0, 1, \dots, m\}$ , we write  $\mathcal{P}^{(h)}$  for the set of rules whose head predicate has stratum  $h$ .

**Example 3.2.** For the program  $\mathcal{P}$  of Example 3.1, the subprograms  $\mathcal{P}^{(0)}$  and  $\mathcal{P}^{(1)}$  are as follows:

$$\begin{aligned} \mathcal{P}^{(0)} &= \left\{ \begin{array}{l} \textit{Happy}(x) \leftarrow \textit{Owns}(x, \textit{iPad}), \textit{Owns}(x, \textit{iPod}) \\ \textit{Happy}(x) \leftarrow \textit{Knows}(x, y), \textit{Happy}(y) \\ \textit{Happy}(x) \leftarrow \textit{Knows}(x, y), \neg \textit{Knows}(x, \textit{Don}) \end{array} \right. \\ \mathcal{P}^{(1)} &= \left\{ \textit{Unhappy}(x) \leftarrow \textit{Owns}(x, y), \neg \textit{Happy}(x) \right. \end{aligned}$$

□

The datalog program is then executed in the following order:

- Execute  $\mathcal{P}^{(0)}$ .
- Execute  $\mathcal{P}^{(1)}$ .
- Execute  $\mathcal{P}^{(2)}$ .
- ...
- Execute  $\mathcal{P}^{(m)}$ .

When we say “Execute  $\mathcal{P}^{(h)}$ ,” we mean to repeatedly instantiate all rules of  $\mathcal{P}^{(h)}$  in all possible ways. That is, repeatedly compute ground rules by replacing variables with constants. Whenever the body of such a ground rule is true, infer that its rule head is also true. When the body of a rule in  $\mathcal{P}^{(h)}$  contains a negated atom  $\neg R_i(\vec{t}_i)$ , then the Stratification Requirement tells us that  $R_i$  is either an IDB predicate with stratum  $< h$ , or an EDB predicate. Therefore, when we execute  $\mathcal{P}^{(h)}$ , all rules that can compute  $R_i$ -facts have already been executed, and we can safely say that a fact  $\neg R_i(\vec{c})$  is true if  $R_i(\vec{c})$  has not been inferred. In other words, when executing  $\mathcal{P}^{(h)}$ , we can treat negated predicates as if they were EDB predicates. We will have more to say about program execution in Section 6.

Note that the rules of a stratified datalog program can be listed in any order. The stratification is the responsibility of the *datalog engine* that executes the program.

**Example 3.3.** The following program computes the complement of the transitive closure of *Knows*.

$$\begin{aligned} \textit{Person}(x) &\leftarrow \textit{Knows}(x, y) \\ \textit{Person}(y) &\leftarrow \textit{Knows}(x, y) \\ \textit{TransitivelyKnows}(x, y) &\leftarrow \textit{Knows}(x, y) \\ \textit{TransitivelyKnows}(x, y) &\leftarrow \textit{Knows}(x, z), \textit{TransitivelyKnows}(z, y) \\ \textit{NotTransitivelyKnows}(x, y) &\leftarrow \textit{Person}(x), \textit{Person}(y), \neg \textit{TransitivelyKnows}(x, y) \end{aligned}$$

Let the strata of *Person*, *TransitivelyKnows*, and *NotTransitivelyKnows* be  $p$ ,  $t$ , and  $n$  respectively. The Stratification Requirement requires  $p \leq n$  and  $t < n$ . These inequalities are satisfied by the following stratification:

stratum 0: *Person*, *TransitivelyKnows*  
stratum 1: *NotTransitivelyKnows*

This stratification gives the following programs  $\mathcal{P}^{(0)}$  and  $\mathcal{P}^{(1)}$ :

$$\begin{aligned}
\mathcal{P}^{(0)} &= \left\{ \begin{array}{l} Person(x) \leftarrow Knows(x, y) \\ Person(y) \leftarrow Knows(x, y) \\ TransitivityKnows(x, y) \leftarrow Knows(x, y) \\ TransitivityKnows(x, y) \leftarrow Knows(x, z), TransitivityKnows(z, y) \end{array} \right. \\
\mathcal{P}^{(1)} &= \{ NotTransitivityKnows(x, y) \leftarrow Person(x), Person(y), \neg TransitivityKnows(x, y) \}
\end{aligned}$$

Therefore, a datalog engine will first execute the rules of  $\mathcal{P}^{(0)}$  until no more facts can be inferred. Then, and only then,  $\mathcal{P}^{(1)}$  is executed. Therefore, when  $\mathcal{P}^{(1)}$  is executed, all *TransitivityKnows*-facts have already been computed, and therefore we know which *TransitivityKnows*-facts are not true.

Note incidentally that a program can have more than one valid stratification. Here is another stratification for this program:

$$\begin{aligned}
\text{stratum 0: } & \textit{TransitivityKnows} \\
\text{stratum 1: } & \textit{Person}, \textit{NotTransitivityKnows}
\end{aligned}$$

It can be shown that the output of a stratified datalog program does not depend on the stratification that is chosen to execute the program: all stratifications result in the same output.  $\square$

To conclude, every stratified datalog program can be regarded as a sequence of smaller datalog programs, one for each stratum, that only use “classical” negation.

**Example 3.4.** A bus company called *Red Cy* uses a binary EDB predicate *Red/2* to store its direct bus connections. Moreover, a predicate *RedCanceled/2* stores the connections that are currently canceled due to a virus outbreak. For example,  $I = \{Red(\text{Mons}, \text{Ath}), Red(\text{Ath}, \text{Dour}), Red(\text{Dour}, \text{Mons}), Red(\text{Mons}, \text{Huy}), Red(\text{Ans}, \text{Mons}), Red(\text{Huy}, \text{Ans}), Red(\text{Ans}, \text{Spa}), Red(\text{Spa}, \text{Huy}), RedCanceled(\text{Ans}, \text{Mons})\}$ . We want to define an IDB predicate *CanAlwaysReturn/1* such that *CanAlwaysReturn(x)* holds true if on any journey that starts from  $x$ , one can always return to  $x$ . For our example database instance, *CanAlwaysReturn(Mons)* is false, because when one travels from *Mons* to *Huy*, one cannot return to *Mons*. On the other hand, *CanAlwaysReturn(Huy)* is true: it is possible to return to *Huy* from every station currently reachable from *Huy*.

When computing a set in datalog, it is sometimes easier to first compute the complement of that set. Let us introduce *CannotAlwaysReturn/1* with the meaning that *CannotAlwaysReturn(x)* is false if and only if *CanAlwaysReturn(x)* is true. So our program will contain the following rules:

$$\begin{aligned}
CanAlwaysReturn(x) &\leftarrow Station(x), \neg CannotAlwaysReturn(x) \\
Station(x) &\leftarrow Red(x, y) \\
Station(y) &\leftarrow Red(x, y)
\end{aligned}$$

The IDB predicate *Station* is useful for making the first rule safe. We are now ready to give the rules for *CannotAlwaysReturn(x)*, which is true if one can travel from  $x$  to some station  $y$ , but there is no journey from  $y$  back to  $x$ .

$$\begin{aligned}
RedTrip(x, y) &\leftarrow Red(x, y), \neg RedCanceled(x, y) \\
RedTrip(x, y) &\leftarrow Red(x, z), RedTrip(z, y), \neg RedCanceled(x, z) \\
CannotAlwaysReturn(x) &\leftarrow RedTrip(x, y), \neg RedTrip(y, x)
\end{aligned}$$

A stratification of this program is as follows:

$$\begin{aligned}
\text{stratum 0: } & \textit{Station}, \textit{RedTrip} \\
\text{stratum 1: } & \textit{CannotAlwaysReturn} \\
\text{stratum 2: } & \textit{CanAlwaysReturn}
\end{aligned}$$

This stratification gives the following programs  $\mathcal{P}^{(0)}$ ,  $\mathcal{P}^{(1)}$ , and  $\mathcal{P}^{(2)}$ :


$$\begin{aligned}
\mathcal{P}^{(0)} &= \left\{ \begin{array}{l} Station(x) \leftarrow Red(x, y) \\ Station(y) \leftarrow Red(x, y) \\ RedTrip(x, y) \leftarrow Red(x, y), \neg RedCanceled(x, y) \\ RedTrip(x, y) \leftarrow Red(x, z), RedTrip(z, y), \neg RedCanceled(x, z) \end{array} \right. \\
\mathcal{P}^{(1)} &= \{ CannotAlwaysReturn(x) \leftarrow RedTrip(x, y), \neg RedTrip(y, x) \} \\
\mathcal{P}^{(2)} &= \{ CanAlwaysReturn(x) \leftarrow Station(x), \neg CannotAlwaysReturn(x) \}
\end{aligned}$$

Nonetheless, programmers can write the rules of their program in any order, and the output is independent of the chosen order.  $\square$

## 4 Program Dependency Graph

The *program dependency graph (PDG)* of a datalog program is a directed edge-labeled graph whose vertices are the IDB predicates of the program. There is a directed edge from  $H$  to  $P$  if there is a rule such that  $H$  is the head predicate, and  $P$  is an IDB predicate that occurs in the body; such an edge is labeled  $-$  if  $P$  occurs within the scope of  $\neg$ .

For example, for the program of Example 3.4, the PDG is as follows:



It is easy to show the following property:

A program  $\mathcal{P}$  is stratified if and only if its PDG has no directed cycle that contains an edge with label  $-$ .

Consequently, in the PDG of a stratified datalog program, any directed (possibly infinite) path that starts from some vertex can traverse at most finitely many edges with label  $-$ . It is then easy to show the following:

Let  $f$  be the function that assigns to each IDB predicate  $P$  the greatest number of edges with label  $-$  on any directed path that starts from  $P$ . We have just argued that this number is finite. Then,  $f$  is a valid stratification. For example, in the preceding PDG,  $f$  assigns 2 to  $CanAlwaysReturn$ , 1 to  $CannotAlwaysReturn$ , and 0 to both  $RedTrip$  and  $Station$ .

## 5 Datalog Languages

In the literature, the name “*datalog*” is usually used only for programs that do not contain negation. For programs with stratified negation, one commonly uses “*stratified datalog*” or “*datalog with stratified negation*.”

“*Datalog with  $\neq$* ” refers to datalog without negation, but with  $\neq$ . In “datalog with stratified negation,” we can compute a *NotEqual/2* predicate, and therefore “datalog with stratified negation and  $\neq$ ” is not more powerful than “datalog with stratified negation.” For example,

$$\begin{aligned} Equal(x, x) &\leftarrow Person(x) \\ NotEqual(x, y) &\leftarrow Person(x), Person(y), \neg Equal(x, y) \end{aligned}$$

A (stratified) datalog program is called *linear* if in the body of each rule, there is at most one occurrence of an IDB predicate that has the same stratum as the rule head.<sup>1</sup> For example, the following rule cannot

<sup>1</sup>Note that, by the Stratification Requirement, such an IDB predicate cannot occur within the scope of  $\neg$  in the rule body.

occur in a linear datalog program:

$$\text{TransitivelyKnows}(x, y) \leftarrow \text{TransitivelyKnows}(x, z), \text{TransitivelyKnows}(z, y). \quad (2)$$

A linear datalog program can contain the following rule:

$$\text{TransitivelyKnows}(x, y) \leftarrow \text{Knows}(x, z), \text{TransitivelyKnows}(z, y). \quad (3)$$

It is generally a good idea to solve a problem by a linear datalog program whenever this is possible, because such programs can be executed more efficiently than nonlinear programs. Informally, (2) is less efficient than (3) because it uses two recursive calls.

A program is in *semi-positive datalog* if negation only applies to EDB predicates.

## 6 Analysis of Program Execution

In Section 3.3, we pointed out that a program  $\mathcal{P}$  is executed by successively executing  $\mathcal{P}^{(0)}, \mathcal{P}^{(1)}, \mathcal{P}^{(2)}, \dots$ . In this section, we discuss in more detail the execution of these subprograms. In particular, for any fixed stratified datalog program  $\mathcal{P}$ , we will discuss the complexity of the following computational problem:

**INPUT:** A database instance.

**OUTPUT:** All IDB facts that can be inferred by  $\mathcal{P}$ .

We will show that this problem is in polynomial time in the size of the input. Note that  $\mathcal{P}$  itself is not part of the input.

### 6.1 Guiding Example

We start by an example explaining the execution of  $\mathcal{P}^{(0)}$  in Example 3.4. This execution is captured by the following pseudo-code:

```

Station0 := ∅
RedTrip0 := ∅
i := 0
loop
  Stationi+1 := {x | ∃y Red(x, y)}
  Stationi+1 := {y | ∃x Red(x, y)}
  RedTripi+1 := {x, y | Red(x, y) ∧ ¬RedCanceled(x, y)}
  RedTripi+1 := {x, y | ∃z (Red(x, z) ∧ RedTripi(z, y) ∧ ¬RedCanceled(x, z))}
  i := i + 1
until a fixed point is reached

```

This pseudo-code initializes the IDB predicates as empty. The loop then repeatedly assigns new values to the IDB predicates. The queries at the right-hand of  $:=$  are relational calculus queries, which can be implemented in relational algebra or SQL, and executed in polynomial time (and even in logarithmic space). For example, the query

$$\{x, y \mid \exists z (\text{Red}(x, z) \wedge \text{RedTrip}^i(z, y) \wedge \neg \text{RedCanceled}(x, z))\} \quad (4)$$

is equivalent to (using numbers for attributes):

```

SELECT  Red.1, RedTripi.2
FROM    Red, RedTripi
WHERE   Red.2 = RedTripi.1
AND     NOT EXISTS ( SELECT *
                     FROM   RedCanceled
                     WHERE  RedCanceled.1 = Red.1
                     AND    RedCanceled.2 = Red.2 )

```

It is easy to see:

- $\text{RedTrip}^0 \subseteq \text{RedTrip}^1$ ; and

<i>Red</i>	1	2
	Mons	Ath
	Ath	Dour
	Dour	Mons
	Mons	Huy
	Ans	Mons
	Huy	Ans
	Ans	Spa
	Spa	Huy

<i>RedCanceled</i>	1	2
	Ans	Mons

<i>Station</i> <sup>0</sup>	1
-----------------------------	---

<i>RedTrip</i> <sup>0</sup>	1	2
-----------------------------	---	---

<i>Station</i> <sup>1</sup>	1
	Mons
	Ath
	Dour
	Huy
	Ans
	Spa

<i>RedTrip</i> <sup>1</sup>	1	2
	Mons	Ath
	Ath	Dour
	Dour	Mons
	Mons	Huy
	Huy	Ans
	Ans	Spa
	Spa	Huy

<i>Station</i> <sup>2</sup>	1
	Mons
	Ath
	Dour
	Huy
	Ans
	Spa

<i>RedTrip</i> <sup>2</sup>	1	2
	Mons	Ath
	Ath	Dour
	Dour	Mons
	Mons	Huy
	Huy	Ans
	Ans	Spa
	Spa	Huy
	Mons	Dour
	Ath	Mons
	Dour	Ath
	Dour	Huy
	Mons	Ans
	Huy	Spa
	Ans	Huy
	Spa	Ans

⋮
---

⋮
---

Figure 1: First two iterations in a fixed point computation.



- if  $RedTrip^i \subseteq RedTrip^{i+1}$ , then  $RedTrip^{i+1} \subseteq RedTrip^{i+2}$ .

Informally, the second item is true because in (4),  $RedTrip^i$  does not occur within the scope of a negation. From these two items, it follows:

$$RedTrip^0 \subseteq RedTrip^1 \subseteq RedTrip^2 \subseteq RedTrip^3 \subseteq \dots$$

The first steps of this computation are shown in Figure 1. Since there are finitely many stations, this sequence must necessarily reach a fixed point. Note that the sequence for *Station* already reaches a fixed point after one iteration.

We now argue that a fixed point will be reached after a number of steps that is polynomial in the size of the database instance. To this end, assume that the database instance has size  $\ell$ . Then, there can be at most  $\ell$  distinct stations, and at most  $\ell^2$  distinct *RedTrip*-facts. Therefore, it must be the case that  $RedTrip^{\ell^2} = RedTrip^{\ell^2+1}$ .

## 6.2 General Treatment

We briefly sketch how the treatment of Section 6.1 generalizes to some  $\mathcal{P}^{(h)}$ , where  $h$  is any stratum. A pseudo-code can be constructed as in Section 6.1, as follows. Assume that  $\mathcal{P}^{(h)}$  contains a rule

$$H(\vec{t}_0) \leftarrow R_1(\vec{t}_1), \dots, R_\ell(\vec{t}_\ell), \neg R_{\ell+1}(\vec{t}_{\ell+1}), \dots, \neg R_m(\vec{t}_m),$$

where the IDB predicates of stratum  $h$  are  $H, R_1, R_2, \dots, R_k$  ( $1 \leq k \leq \ell$ ), while  $R_{k+1}, \dots, R_m$  are either IDB predicates of a smaller stratum or EDB predicates. In particular,  $R_{\ell+1}, \dots, R_m$  cannot be IDB predicates of stratum  $h$  because of the Stratification Requirement. Then, the **loop** of the pseudo-code contains an instruction:

$$H^{i+1} := \left\{ \vec{t}_0 \mid \exists \vec{x} \left( \begin{array}{l} R_1^i(\vec{t}_1) \wedge \dots \wedge R_k^i(\vec{t}_k) \wedge \\ R_{k+1}(\vec{t}_{k+1}) \wedge \dots \wedge R_\ell(\vec{t}_\ell) \wedge \neg R_{\ell+1}(\vec{t}_{\ell+1}) \wedge \dots \wedge \neg R_m(\vec{t}_m) \end{array} \right) \right\}, \quad (5)$$

where  $\vec{x}$  contains the variables that are not in  $\vec{t}_0$ .

Since the IDB predicates of stratum  $h$  (i.e.,  $R_1^i, \dots, R_k^i$  in the query (5)) do not occur within the scope of a negation, the pseudo-code must necessarily reach a fixed point. After how many steps will this fixed point be reached? Let  $n$  be the arity of  $H$ , and assume a database instance of size  $\ell$ . Since such a database instance can contain at most  $\ell$  constants, we can generate at most  $\ell^n$  distinct  $H$ -facts from this database, which is a polynomial number (because  $n$  is fixed). It follows that a fixed point will be reached in polynomial time in the size of the database. To conclude, the program  $\mathcal{P}^{(h)}$  runs in polynomial time. Since the number of strata is constant (it is not part of the input), every stratified datalog program can be evaluated in polynomial time.

## 7 Datalog Engine

A datalog engine called DLV can be downloaded at <http://www.dlvsystem.com/>. To run the program of Example 3.4, we create the text file `br01.txt` as shown in Figure 2. Constants start with a lowercase letter, and variables with an uppercase letter. More information on the DLV syntax can be found at [http://www.dlvsystem.com/html/DLV\\_User\\_Manual.html#AEN928](http://www.dlvsystem.com/html/DLV_User_Manual.html#AEN928). Note that DLV supports a larger language than stratified datalog with  $\neq$ . For example, DLV accepts programs in which negation is not stratified. However, in the course *Bases de Données II*, the aim is to use only stratified datalog. When we execute our program by means of the command

```
dlv -filter=CanAlwaysReturn br01.txt
```

we get the following output:

```
{CanAlwaysReturn(huy), CanAlwaysReturn(ans), CanAlwaysReturn(spa)}
```

For help, use the command `dlv -help`.

A common situation is where we want to separate the database from the datalog program, so that the same database can be used with different programs. The following command executes the program `pgm.txt` on the database in `db.txt`:

```
dlv db.txt pgm.txt
```

```

% This is file bro1.txt
% The database.
Red(mons, ath).
Red(ath, dour).
Red(dour, mons).
Red(mons, huy).
Red(ans, mons).
Red(huy, ans).
Red(ans, spa).
Red(spa, huy).
RedCanceled(ans, mons).

% The program.
CanAlwaysReturn(X) :- Station(X), not CannotAlwaysReturn(X).
Station(X) :- Red(X,Y).
Station(Y) :- Red(X,Y).
Redtrip(X,Y) :- Red(X,Y), not RedCanceled(X,Y).
Redtrip(X,Y) :- Red(X,Z), Redtrip(Z,Y), not RedCanceled(X,Z).
CannotAlwaysReturn(X) :- Redtrip(X,Y), not Redtrip(Y,X).

```

Figure 2: The program of Example 3.4 in DLV syntax.